FILE PW21.

INTERNAL DOCUMENT 234

I.O.S.

PITCH ROLL BUOY DIRECTIONAL WAVE DATA PROCESSING USING AN ACORN MICROCOMPUTER

HARDWARE & SOFTWARE DETAILS

K.G.Birch and J.A.Ewing

March 1985

NATURAL ENVIRONMENT RESEARCH COUNCIL
INSTITUTE OF OCEANOGRAPHIC SCIENCES

INSTITUTE OF OCEANOGRAPHIC SCIENCES

Wormley, Godalming,
Surrey GU8 5UB
(042-879-4141)

(Director: Dr. A. S. Laughton, FRS)

Bidston Observatory,
Birkenhead,
Merseyside L43 7RA
(051-653-8633)
(Assistant Director: Dr. D. E. Cartwright)

Crossway,
Taunton,
Somerset TA1 2DW
(0823-86211)
(Assistant Director: M. J. Tucker)

PITCH ROLL BUOY DIRECTIONAL WAVE DATA PROCESSING USING AN ACORN MICROCOMPUTER

HARDWARE & SOFTWARE DETAILS

K.G.Birch and J.A.Ewing

March 1985

INTERNAL DOCUMENT NO.234

INSTITUTE OF OCEANOGRAPHIC SCIENCES
WORMLEY,GODALMING
SURREY GU8 5UB.

Contents

Circuits and Diagrams

1. Objective.

In many of today´s wave measurements programmes it is desirable to produce real time directional wave spectra, a system using an inexpensive microcomputer has been developed to accommodate this need.

2. Introduction

The pitch roll buoy is regularly used for the collection of wave data in the open ocean from an attendant ship.Normally the data is recorded in digital format on magnetic tape and after the cruise it is processed on a mainframe computer.This system works well but the processed data is not available until some time after the cruise.The laboratory mainframe computer has one advantage over the microcomputer that of speed,however this is not an overriding consideration at sea as sequential wave measurements are at least one hour apart and in routine sampling schemes measurements are usually separated by three hours.

The implementation of near real time processing is achieved by transmitting the serial data from the Pitch Roll Buoy Monitor Unit asynchronously to the microcomputer.The data is stored on disc in its raw format,and subsequently processed using the Fast Fourier Transform (FFT) algorithm to produce a directional spectrum with the results being stored on disc.The spectra are then plotted on linear or logarithmic axes to VDU and/or printer.The availabilty of near real time wave information thus allows decisions to be made on the planning and organization of experimental programs of work at sea.

3. System Description.

Data Acquisition.

The microcomputer used in the system is the Acorn BBC Model B,with 6502 Second Processor,Dual Disc drive and Epson FX-80 Printer.The microcomputers memory is insufficient to hold the raw data set and carry out the processing ,therefore the data is written to disc as it is acquired.The Operating System requires the use of the resident processor to write data from the temporary storage buffer to the disc controller,this means that although the software can be synchronised directly to the Pitch Roll Buoy any disc write operation will lose data and synchronisation.To overcome this problem it is necessary to allow the data transmission to between Buoy and Computer to be asynchronous, the use of a First in First Out Register (FIFO) makes this possible.

The data format used in the Pitch Roll Buoy is ten channels of three digit Binary Coded Decimal (BCD).The data from channels 0-3 (Acceleration,Pitch,Roll and Compass) is repeated in channels 4-7 with channel 8 as the record number and channel 9 as the scan count.To reduce the data storage to a minimum only the first four channels of data are stored with redundant data being rejected by software.The hardware interface (fig.1) causes the BCD digits to be strobed into FIFO with one character in the least significant nibble of each byte (thirty characters per data frame).At the end of each data frame an extra strobe is generated,with the most significant nibble of the data byte being forced true.

The transfer of data between the FIFO and the User Port is carried out under software control with a handshake routine.At the start of data acquisition the program prompts the operator to CLEAR THE MICROBUFFER (FIFO) this deletes any data previously acquired.The data is then entered into the microcomputer when the CB1 line of the User Port is strobed by the FIFO.When the strobe is received the User port latch is read and the byte is checked to determine if the most significant nibble is true,if false the byte is discarded.The User Port then strobes the FIFO using the CB2 line.The FIFO then presents the next data byte to the User Port and strobes the CB1 line.This sequence of acquiring bytes continues until an end of frame flag is detected,this establishes software synchronisation with the knowledge that the next byte will be the first data character of the next data frame.The data bytes are then acquired using the handshake routine,each being checked for end of frame flag,if false the byte is stored in memory until an end of frame flag is detected.The three BCD characters of each channel are then used to calculate the channel value as an integer variable within the computer.All ten channels are then displayed on the VDU and the first four channels are written to disc.The computer continues to acquire data until the required number of data points is reached.The ability of the software to recognise the end of data frame by testing for the high nibble being true means that any errors within any one frame of data do not cause errors in subsequent data frames.

The Operating System of the microcomputer stores data to be written to disc in a 256 byte buffer.When the buffer is full the Operating System halts the program and passes the data through the processor to the disc controller.Each time a disc write operation is made the disc must first be spun up to speed,this combined with the actual write operation takes approximately two seconds,and it is this delay that makes input of data under software control mandatory.Despite this delay the computer is easily able to catch up time lost during the disc write cycle.

Data Processing

Theory

The method of computing the directional wave spectrum from the pitch-roll buoy follows the theory given by Longuet-Higgins et al;(1963: In "Ocean Wave Spectra",Prentice-Hall, 111-136).Essentially the directional spectrum is derived from the six cross spectral estimates obtained from correlating the heave acceleration with the two component slopes in the North-East directions.We denote heave acceleration by the subscript 1 and the slopes in north and east directions by the subscripts 2 and 3 respectively.

The basic data thus consists of the cross-spectra C11,C22,C33,Q12,Q13 and C23.From these data,estimates of the normalized angular harmonics A1,B1,A2,B2 and the check ratio,R,are obtained.Finally a number of statistical parameters are evaluated,namely:-

TH1 = mean wave direction from 1st order angular harmonics,
TH12 = mean wave direction from 2nd order angular harmonics,
TH2 = directional spread,
S1,S2 = estimates of exponent of cosine-power law obtained from 1st and 2nd order angular harmonics.

Method of Processing

The processing proceeds in three stages:-
(i)Program GEO reads the raw data of heave acceleration,pitch,roll and compass and computes geo-orientated data in the form of heave acceleration and two component slopes(That is,the series 1,2 and 3 referred to above).These data are written to disc.Each series consists of 4 groups of 1024 data points sampled simultaneously at 0.5 sec interval.CAL2 and CAL3 are calibrations for pitch and roll in units of radians per digital unit(d.u.) CAL 4 is the compass calibration(radians per d.u.),OF4 is the magnetic variation offset(in radians).

(ii)Program DWS carries out the main computation.The appropriate cross-spectra(see above) are evaluated in turn by reading the data from selected pairs of channels previously stored on disc using program GEO .Spectral estimates are obtained using the Fast Fourier Transform (FFT) algorithm over 1024 data points and repeating this operation over four sections.Finally,spectral estimates are formed from the average over seven adjacent harmonics,thus having fifty six degrees of freedom with frequencies at 7/512 Hz interval.
CAL1 is the calibration for heave in units of ´g´ per d.u. Procedure DPAR calculates the statistical parameters.Procedure FFT is the Fast Fourier Transform.
The cross spectra,angular harmonics,check ratio and directional parameters are printed out.Finally the wave height spectrum,mean direction and angular spread are written to disk ready for plotting.

(iii) Program PLMZ reads the data from disc and plots to VDU a selected parameter under operator command from the User Keys.The wave height spectrum can be plotted on linear axes or logarithmic axes.If linear axes are chosen a maximum scale value of either 0.1,1.0 or 10.0 M*M/Hz is selected to to give the best screen resolution of the data.
The directional parameters TH1 and TH2 are plotted together on linear axes.
Hard copy of any graphical display on the screen may be obtained via User Key menu.

4.Program Timing

The time taken for the analysis of four sections of data (each of 1024 records) is thirty minutes, this time will be reduced in the future by the use of an FFT Rom.

SOFTWARE DESCRIPTION

Data Acquisition

PRBDW

Lines 10-90	Main program:This is three Procedures,the first two (PROCsetup and PROCscreen) control screen headings and text windows.The third procedure (PROCdin) controls the input of data,the writing to screen and the storing of data to disc.
Lines 110-160	PROCsetup:This procedure stores the screen heading as a series of strings which are accessed by PROCscreen.
Lines 180-250	PROCscreen:This procedure writes the screen heading and sets up the text window.
Lines 260-300	PROCdin.Clears screen and prompts the operator to enter file name.Drive 1 is enabled and a fileopened with the file name written to disc.
Lines 310-340	Clears screen and writes command to ´CLEAR MICROBUFFER´ to screen and awaits command input to start data acquisition.
Lines 360-400	Declares addresses used by labels in assembler routines.Dimensions storage area for machine code.
Lines 440-490	Initialises storage pointers and counters.
Lines 500-720	RUPT is an interrupt routine which is executed immeadiatly the FIFO strobes CB1.The current contents of the accumulator,the X and the Y registers are saved to stack and recalled at completion of the routne.The input port is read and saved in a temporary storage area,the high nibble is checked for end of frame flag,which if true causes the end of frame marker to be set and no more bytes are called.However if the high nibble is false the byte is stored in memory until a complete data set has been collected.After each data byte is stored the CB2 line is strobed with a 6 microsecond pulse and the bytes which were saved to stack are reloaded to the accumulator,and to the X and Y registers.For the duration of the CB2 pulse all interrupts are inhibited ensuring the length of the pulse is not affected at any time by a system interrupt.When a frame flag is detected in the data,a marker (FF hex) is set which the Basic program waiting to detect.
Lines 730,840	The routine GO sets up the User Port for the input of data latched by a negative going pulse on CB1,and the IRQ2 vectors are set to point to the routine RUPT.
Lines 850-930	The routine STRB is called after the data set has been processed,it resets the end of frame marker and strobes the CB2 line to commence the input of a new data frame.
Lines 950-960	Calls the initialisation routines.
Lines 970-980	Sets print format and sets data frame counter to zero.
Lines 990-1000	Causes strobe to FIFO to be generated and waits for data frame marker to be detected.
Line 1010	Detects first data frame marker and causes data prior to that time to be ignored.

Lines 1020-1060 Prints all ten channels of data to screen and writes first four channels to disc.

Lines 1070-1090 Print clear line on screen to make data easier to read, increments frame counter and checks whether the record is completed.

Lines 1100-1110 File closed and drive 0 enabled.

PRBDW

```
 10 REM PROG IS CALLED PRBDW
 20 ************************
 30 REM THIS PROG INPUTS DATA FORM P/R BUOY
 40 REM FIRST FOUR DATA CHANNELS ARE WRITTEN TO DISC
 50 MODE 7
 60 PROCsetup
 70 PROCscreen
 80 PROCdin
 90 END
100 *********************
110 DEFPROCsetup
120 DIM TITLE$(2)
130 TITLE$(0)="              PITCH ROLL BUOY"
140 TITLE$(1)=" ACC     ROLL     ACC     ROLL     R NUM"
150 TITLE$(2)="    PITCH  COMPASS  PITCH  COMPASS   SCAN"
160 ENDPROC
170 *********************
180 DEF PROCscreen
190 FOR N%=0 TO 2
200 IF N%=1 THEN PRINT
210 PRINTTITLE$(N%)
220 NEXT
230 VDU28,0,23,39,4
240 ENDPROC
250 *********************
260 DEFPROCdin
270 CLS
280 *DRIVE 1
300 INPUTTAB(10,10)"INPUT FILE NAME",A$
310 X=OPENOUT A$:PRINT£X,A$
315 CLS
320 PRINTTAB(10,6)"CLEAR MICROBUFFER"
330 PRINTTAB(2,9)"PRESS ANY KEY TO START DATA INPUT"
340 B$=GET$
350 ?&8E=0
360 IRQ2V=&0206
370 OSASCI=&FFE3
380 OSBYTE=&FFF4
390 STORE =&3000
400 DIM INIT 400
410 FOR I%= 0 TO 2 STEP 2
420 P%=INIT
430 [OPT I%
440.INIT  LDA £&00
450       STA &8F
460       STA &8E                    \RESET POINTERS
470       LDA £&E0
480       STA &FE6C                  \SET CB2 HIGH
490       RTS
500.RUPT  PHA:TXA:PHA:TYA:PHA        \SAVE REGISTERS
510       LDA &FE60                  \READ INPUT PORT
520       STA &301F                  \SAVE IN TEMP STORE
530       AND £&F0                   \MASK LOW 4 BITS
540       CMP £&F0                   \CHECK FOR FRAME SYNC
```

```
  550        BEQ FRAME                    \DETECT FRAME
  560        LDA &8E                      \LOAD STORAGE POINTER
  570        TAX                          \STORE IN X REG
  580        LDA &301F                    \RELOAD DATA
  590        AND £&0F                     \MASK OFF HIGH 4 BITS
  600        STA STORE,X                  \STORE DATA
  610        INC &8E                      \INCREMENT STORAGE POINTER
  620        LDA £&C0
  630        SEI                          \DISABLE INTERRUPTS
  640        STA &FE6C                    \SET CB2 LOW
  650        LDA £&E0
  660        STA &FE6C                    \SET CB2 HIGH
  670        CLI                          \ENABLE INTERRUPTS
  680        JMP FINAL
  690.FRAME LDA £&FF                      \SET FRAME SYNC PTR
  700        STA &8E                      \STORE STORAGE POINTER
  710.FINAL PLA:TAY:PLA:TAX:PLA           \RESTORE REGISTERS
  720        RTI
  730.GO     SEI                          \DISABLE INTERRUPTS
  740LDA £&97:LDX £&6C                    \SELECT PCR ON USER VIA
  750        LDY £0:JSR OSBYTE            \SET UP TO DETECT H-L TRANS.

  760        LDA £&97:LDX £&6E            \SELECT IER ON USER VIA
  770        LDY £&6F:JSR OSBYTE          \DISABLE ALL BUT CB1
INTERRUPTS
  780        LDA £&97:LDX £&6E            \SELECT IER ON USER VIA
  790        LDY £&90:JSR OSBYTE          \ENABLE CB1 INTERRUPTS
  800                                     \SET IRQ2 VECTORS TO POINT TO
RUPT
  810        LDA £RUPT MOD 256:STA IRQ2V
  820        LDA £RUPT DIV 256:STA IRQ2V+1
  830        CLI                          \ENABLE INTERRUPTS
  840        RTS
  850.STRB   LDA £&00
  860        STA &8E                      \RESET FRAME SYNC
  870        LDA £&C0
  880        SEI                          \DISABLE INTERRUPTS
  890        STA &FE6C                    \SET CB2 LOW
  900        LDA £&E0
  910        STA &FE6C                    \SET CB2 HIGH
  920        CLI                          \ENABLE INTERRUPTS
  930        RTS
  940]NEXT I%
  950 CALL INIT
  960 CALL GO
  970 @%=04
  980 J=0
  990 CALL STRB
 1000 IF ?&8E=255 THEN 1010 ELSE 1000
 1010 IF J=0 THEN 1080
 1020 FOR I=12288 TO 12317 STEP 3
 1030 D%=INT(?I*100+?(I+1)*10+?(I+2))
 1040 PRINT D%;
 1050 IF I<12298 THEN PRINT£X,D%
 1060 NEXT
 1070 PRINT
 1080 J=J+1
 1090 IF J=4098 THEN 1100 ELSE 990
```

```
1100 CLOSE£X
1110 *DRIVE0
1120 ENDPROC
1130 ********************
```

Data Processing

GEO

Lines 10-60	Rem statements with program description,also array dimensions and output format code.
Lines 70-130	Screen description of disc configuration required for data processing.
Lines 140-180	Calibration values.
Lines 190-220	File name entry via keyboard.
Lines 240-280	Open input and output files ready for processing.
Lines 290-300	Input File name of raw data
Lines 310-330	Loop for reading data, selecting drive 1 for data input and setting pitch/roll total variables to zero.
Lines 340-380	Read 1024 data points and sums pitch and roll values.
Lines 390-410	Calculates mean value of pitch and roll.
Lines 420-560	Calculates the slopes from pitch and roll values.The acceleration and slope values are then written to disc.
Lines 570-580	Prints end of section data points to screen,then returns for next section.
Lines 590-660	Close files,write instructions to screen as to how to continue processing.

GEO

```
   10 REM PROGRAM TO READ P-R DATA AND COMPUTE COMPONENT
SLOPES.FINALLY WRITING TO DISK READY FOR CROSS-SPECTRAL ANALYSIS
   20 DIM A(1024),P(1024),R(1024),C(1024)
   30 REM A,P,R & C ARE HEAVE ACCN,PITCH,ROLL &COMPASS
   40REM MIZEX CALS. IN RADIANS/DIG. UNIT:MAY 1984
   50 @%=&20208
   60 CLS
   70 PRINT TAB(5,7)" RAW DATA DISK IN DRIVE ´1´"
   80 PRINT TAB(5,10)"PLACE EMPTY FORMATTED DISK"
   90 PRINT TAB(12,12)"IN DRIVE ´0´"
  100 PRINT TAB(5,20)"PRESS ANY KEY TO CONTINUE"
  110 V$=GET$
  120 FOR I=0 TO 200
  130 CLS
  140 CAL2=-1.086*10^(-3)
  150 CAL3=1.150*10^(-3)
  160 CAL4=0.02454
  170 REM ASSUME MAG. VARN. IS 14 DEG. WEST
  180 OF4=-0.244
  190 CLS
  200 INPUT TAB(5,8) "FILE NAME WITH RAW DATA",A$
  210 INPUT TAB(5,11) "FILE NAME FOR GEO-SERIES",AA$
  230 REM LOOP 4 TIMES OVER 1024 RECORDS
  240 *DRIVE 0
  250 Z=OPENOUT AA$
  260 PRINT£Z,AA$
  270 *DRIVE 1
  280 ZZ=OPENIN A$
  290 CLS
  300 INPUT£ZZ,C$:PRINT TAB(16,10) C$
  310 FOR J%=1 TO 4
  320 *DRIVE 1
  330 S1=0.:S2=0.
  340 FOR I%=1 TO 1024
  350 INPUT£ZZ,A(I%),P(I%),R(I%),C(I%)
  360 S1=S1+P(I%)
  370 S2=S2+R(I%)
  380 NEXT I%
  390 REM COMPUTE MEANS
  400 MP=S1/1024.
  410 MR=S2/1024.
  420 *DRIVE 0
  430 REM GEO-ORIENTATE SERIES
  440 FOR I%=1 TO 1024
  450 THETA=C(I%)*CAL4+OF4
  460 THETA=0.0
  470 U=COS(THETA)
  480 V=SIN(THETA)
  490 S=P(I%)-MP
  500 T=R(I%)-MR
  510 S=S*CAL2
  520 T=T*CAL3
  530 X=S*U-T*V
```

```
540 Y=T*U+S*V
550 PRINT£Z,A(I%),X,Y
560 NEXT I%
570 PRINT J%,A(1024),X,Y
580 NEXT J%
590 CLOSE£ZZ
600 CLOSE£Z
610 CLS
620 PRINT TAB(8,4)"TO CONTINUE PROCESSING"
630 PRINT TAB(1,8)"TRANSFER DISK CONTAINING FILE ";CHR$(136);AA$
640 PRINT TAB(6,10)"FROM DRIVE ´0´ TO DRIVE ´1´"
650 PRINT TAB(4,12)"PLACE SYSTEMS DISK IN DRIVE ´0´"
660 PRINT TAB(3,18);"TYPE";CHR$(136);"CHAIN DWS::"CHR$(137);"AND
PRESS RETURN"
```

DWS

Lines 10-70	Set up arrays.A and B are used in FFT.Arrays Cxy are cross spectra.Arrays ACxy are averaged cross spectra.
Lines 80-150	Array size is 1024. T is length of section in seconds. NH% is the number of harmonics for spectral average.
Lines 160-230	Set initial cross spectra to zero.
Lines 250-290	Input number of sections to be processed.CAL1 is heave calibration in ´g´ per digital units
Lines 300-420	Set up file names. M% is number of sections
Lines 430-640	Compute cross spectra for series 1 and 2 .Print out last array values for check purposes.
Lines 650-810	Cross spectra for series 1 and 3
Lines 820-980	Cross spectra for series 2 and 3
Lines 990-1190	Compute average cross spectra(over NH% adjacent harmonics).AE is wave height spectrum obtained from C11 in usual method.
Lines 1200-1270	Print out cross-spectral estimates
Lines 1280-1490	Evaluate and print out angular harmonics(A1,B1,A2,B2) check ratio (R) and directional parameters(TH1,TH2). Write AE,TH1,TH2 to disc ready for plotting.
Lines 1500-1760	Procedure DPAR evaluates directional parameters given angular harmonics
Lines 1770-2180	Procedure for Fast Fourier Transform

DWS

```
   10 CLS
   20 INPUT TAB(10,10)"SKIP 0,1,2 OR 3 SECTIONS";SKIP
   30 MODE 7
   40 DIM
A(1024),B(1024),C11(270),C22(270),C33(270),C23(270),Q12(270),Q13(270)
   50 DIM AE(37),AC11(37),AC22(37),AC33(37),AC23(37),AQ12(37),AQ13(37)
   60 REM SET ARRAY SIZE N
   70 N%=1024
   80 *DRIVE 1
   90 T=512.
  100 NH%=7
  110 REM SET RANGE OF FREQUENCIES
  120 NL=3:NU=37
  130 NN%=270
  140 V%=-1
  150 RT=1./T
  160 FOR K%=1 TO 270
  170 C11(K%)=0.
  180 C22(K%)=0.
  190 C33(K%)=0.
  200 C23(K%)=0.
  210 Q12(K%)=0.
  220 Q13(K%)=0.
  230 NEXT K%
  240 INPUT TAB(10,10)"NUMBER OF SECTIONS",M%
  250 CLS
  260 R=0.5*T/(M%*NH%)
  270 CAL1=9.926*10^(-4)
  280 R1=R*CAL1*CAL1
  290 R2=R*CAL1
  300 CLS
  310 INPUT TAB(10,10)"FILE NAME OF INPUT DATA",A$
  320 CLS
  330 INPUT TAB(10,10)"FILE NAME OF OUTPUT DATA",B$
  340 CLS
  350 VDU2
  360 PRINT TAB(5,10)"FILE NAME OF INPUT  ";A$
  370 PRINT TAB(5,13)"FILE NAME OF OUTPUT ";B$
  380 PRINT TAB(10,15),"NO. OF SECTIONS",M%
  390 PRINT "NO. OF HARMONICS IN SPECTRAL ESTIMATES",NH%
  400 PRINT
  410 PRINT "DATA VALUES":PRINT
  420 VDU3
  430 @%=&20208
  440 Z=OPENIN A$
  450 INPUT£Z,C$
  460 ADD=LEN(C$)+2
  470 PTR£Z=18432*SKIP+ADD
  480 FOR L%=1 TO M%
  490 FOR I%=1 TO 1024
  500 INPUT£Z,A(I%),B(I%),DUM
  510 NEXT I%
  520 VDU2:PRINT L%,A(1024),B(1024):VDU3
```

```
530  PROCFFT(10,N%,V%)
540 FOR J%=2 TO NN%
550 Q%=N%-J%+2
560 C1=A(J%)+A(Q%)
570 C2=B(J%)+B(Q%)
580 S1=B(J%)-B(Q%)
590 S2=-A(J%)+A(Q%)
600 C11(J%)=C11(J%)+C1*C1+S1*S1
610 C22(J%)=C22(J%)+C2*C2+S2*S2
620 Q12(J%)=Q12(J%)+S1*C2-C1*S2
630 NEXT J%
640 NEXT L%
650 PTR£Z=18432*SKIP+ADD
660 FOR L%=1 TO M%
670 FOR I%=1 TO 1024
680 INPUT£Z,A(I%),DUM,B(I%)
690 NEXT I%
700 VDU2:PRINT L%,A(1024),B(1024):VDU3
710 PROCFFT(10,N%,V%)
720 FOR J%=2 TO NN%
730 Q%=N%-J%+2
740 C1=A(J%)+A(Q%)
750 C3=B(J%)+B(Q%)
760 S1=B(J%)-B(Q%)
770 S3=-A(J%)+A(Q%)
780 C33(J%)=C33(J%)+C3*C3+S3*S3
790 Q13(J%)=Q13(J%)+S1*C3-C1*S3
800 NEXT J%
810 NEXT L%
820 PTR£Z=18432*SKIP+ADD
830 FOR L%=1 TO M%
840 FOR I%=1 TO 1024
850 INPUT£Z,DUM,A(I%),B(I%)
860 NEXT I%
870 VDU2:PRINT L%,A(1024),B(1024):VDU3
880 PROCFFT(10,N%,V%)
890 FOR J%=2 TO NN%
900 Q%=N%-J%+2
910 C2=A(J%)+A(Q%)
920 C3=B(J%)+B(Q%)
930 S2=B(J%)-B(Q%)
940 S3=-A(J%)+A(Q%)
950 C23(J%)=C23(J%)+C2*C3+S2*S3
960 NEXT J%
970 NEXT L%
980 CLOSE£Z
990 FOR I%=NL TO NU
1000 AE(I%)=0.
1010 AC11(I%)=0.
1020 AC22(I%)=0.
1030 AC33(I%)=0.
1040 AC23(I%)=0.
1050 AQ12(I%)=0.
1060 AQ13(I%)=0.
1070 FOR J%=1 TO NH%
1080 K%=J%+NH%*I%
1090 FREQ=(K%-1)*RT
1100 RF4=96.2/(6.2832*FREQ)^4
```

```
1110 AE(I%)=AE(I%)+RF4*C11(K%)*R1
1120 AC11(I%)=AC11(I%)+C11(K%)*R1
1130 AC22(I%)=AC22(I%)+C22(K%)*R
1140 AC33(I%)=AC33(I%)+C33(K%)*R
1150 AC23(I%)=AC23(I%)+C23(K%)*R
1160 AQ12(I%)=AQ12(I%)+Q12(K%)*R2
1170 AQ13(I%)=AQ13(I%)+Q13(K%)*R2
1180 NEXT J%
1190 NEXT I%
1200 MODE3:VDU2
1210 @%=&0000040A
1220 PRINT "     F        C11        C22        C33        Q12        Q13
   C23"
1230 PRINT
1240 FOR I%= NL TO NU
1250 FREQ=(NH%*I%+2)*RT
1260 PRINT FREQ,AC11(I%),AC22(I%),AC33(I%),AQ12(I%),AQ13(I%),AC23(I%)
1270 NEXT I%
1280 PRINT:PRINT
1290 PRINT "    F     E     A1     B1     A2     B2     R      TH1   TH12
TH2    S1    S2"
1300 PRINT
1310 @%=&01020206
1320 Q=OPENOUT B$
1330 FOR I%=NL TO NU
1340 FREQ=(7*I%+2)*RT
1350 S=AC11(I%)
1360 XX=AC22(I%)+AC33(I%)
1370 X1=SQR(S*XX)
1380 A1=AQ12(I%)/X1
1390 B1=AQ13(I%)/X1
1400 A2=(AC22(I%)-AC33(I%))/XX
1410 B2=2.*AC23(I%)/XX
1420 REX=SQR(S/XX)
1430 PROCDPAR(A1,B1,A2,B2)
1440 PRINT FREQ,AE(I%),A1,B1,A2,B2,REX,TH1,TH12,TH2,SP1,SP2
1450 PRINT£Q,I%,AE(I%),TH1,TH2
1460 NEXT I%
1470 VDU3:MODE 7
1480 CLOSE£Q
1490 END
1500 ****************************
1510 DEF PROCDPAR(A1,B1,A2,B2)
1520 LOCAL X1,X2,Y1,Y2,Y3,Z,Z1,C1,C2
1530 X1=ATN(B1/A1)
1540 Z=DEG(X1)
1550 IF B1>0. AND A1<0. THEN Z=Z+180.
1560 IF B1<0. AND A1<0. THEN Z=Z-180.
1570 IF Z<0. THEN Z=360.+Z
1580 TH1=Z
1590 C1=SQR(A1*A1+B1*B1)
1600 IF C1>1. THEN C1=1.
1610 TH2=SQR(2.-2.*C1)
1620 SP1=2./(TH2*TH2)-1.
1630 TH2=DEG(TH2)
1640 A2=A2+.0001
1650 X2=ATN(B2/A2)
1660 Z1=0.5*DEG(X2)
```

```
1670 IF B2>0. AND A2<0. THEN Z1=Z1+90.
1680 IF B2<0. AND A2<0. THEN Z1=Z1-90.
1690 IF Z1<0. THEN Z1=360.+Z1
1700 TH12=Z1
1710 C2=SQR(A2*A2+B2*B2)
1720 Y1=1.+3.*C2
1730 Y2=SQR(1.+14.*C2+C2*C2)
1740 Y3=2.*(1.-C2)
1750 SP2=(Y1+Y2)/Y3
1760 ENDPROC
1770 ****************************
1780 DEF PROCFFT(N%,B%,V%)
1790 LOCAL L%,D%,K%,M%,ME%,J%,TR,TI
1800 FOR J%=1 TO B%
1810 A(J%)=A(J%)/B%
1820 B(J%)=B(J%)/B%
1830 NEXT
1840 D%=B% DIV 2
1850 G%=B%-1
1860 J%=1
1870 FOR L%=1 TO G%
1880 IF L%>J% OR L%=J% GOTO 1920
1890 TR=A(J%):TI=B(J%)
1900 A(J%)=A(L%):B(J%)=B(L%)
1910 A(L%)=TR:B(L%)=TI
1920 K%=D%
1930 IF K%>J% OR K%=J% GOTO 1970
1940 J%=J%-K%
1950 K%=K% DIV 2
1960 GOTO 1930
1970 J%=J%+K%
1980 NEXT
1990 FOR M%=1 TO N%
2000 UR=1.0:UI=0.0
2010 ME%=2^M%
2020 K%=ME% DIV 2
2030 PK=PI/K%
2040 WR=COS(PK):WI=V%*SIN(PK)
2050 FOR J%=1 TO K%
2060 FOR L%=J% TO B% STEP ME%
2070 Z%=L%+K%
2080 TR=A(Z%)*UR-B(Z%)*UI
2090 TI=A(Z%)*UI+B(Z%)*UR
2100 A(Z%)=A(L%)-TR:B(Z%)=B(L%)-TI
2110 A(L%)=A(L%)+TR:B(L%)=B(L%)+TI
2120 NEXT
2130 U0=UR*WR-UI*WI
2140 UI=UR*WI+UI*WR
2150 UR=U0
2160 NEXT:NEXT
2170 ENDPROC
2180 *****************************
```

PLMZ

Lines 10-20	This configures the function keys ready for plotting to be entered when prompted by screen,also loads screen dump routine into memory.
Lines 30-240	Enters file name of data to be plotted,reads data and sets plotting scale(IRGE).
Lines 250-300	Resets values above 0.5Hz
Lines 310-400	Operator prompted to select via user keys plotting requirements.Error message if incorrect key depressed.Screen dump is not allowed until plot has been performed.
Lines 400-430	Screen initialisation
Lines 440-510	Sets frequency interval and offset.Call PROCBOX to draw graph external lines.Determine using value of II which plotting procedures are required.
Lines 520-570	Call procedures to plot directional parameteres,and enable screen dump.Return to keyboard menu.
Lines 580-620	Call procedures to plot energy spectra on logarithmic axis,enables screen dump.Returns to keyboard menu.
Lines 630-680	Call procedures to plot energy spectra on linear axis,enables screen dump.
Lines 690-710	Call screen dump routine ,then returns to keyboard menu.
Lines 720-730	Returns system to Drive 0, turn off printer,program end.
Lines 750-840	Procedure to plot graph outline.
Lines 860-1230	Procedure for annotation of logarithmic axes.
Lines 1240-1380	Procedure plots data on logarithmic scale.
Lines 1390-1770	Procedure for annotation of linear axes.
Lines 1780-1960	Procedure plots data on linear scale.
Lines 1970-2320	Procedure for annotation of axes for directional plot.
Lines 2330-2570	Procedure for plotting directional data.

PLMZ

```
10 *FX225,200
20 *LOAD"DUMPCL2"
30 *DRIVE 1
40 DIM E(128),TH1(128),TH2(128)
50 REM THIS IS PROGRAM PLMZ
60 REM IT READS 1-D SPECTRUM E FROM DISK
70 REM ROUTINE FOR PLOTTING 1 & 2-D SPECTRA FROM MIZEX
80 REM LINEAR OR LOG PLOTS ARE MADE
90 REM PLOT IS SELECTED USING II
100 MODE 4
110 PRINTTAB(17,8)"FILE NAME "
120 INPUT TAB(17,8)"FILE NAME "A$
130 CLS
140 PRINTTAB(17,8)A$
150 EMAX=0
160 X=OPENIN A$
170 FOR I%=3 TO 37
180 INPUT£X,I%,E(I%),TH1(I%),TH2(I%)
190 IF E(I%)>EMAX THEN EMAX=E(I%)
200 NEXT I%
210 CLOSE£X
220  IRGE=1
230 IF EMAX>0.1 THEN IRGE=2
240 IF EMAX>1.0 THEN IRGE=3
250 REM SET VALUES ABOVE 0.5 HZ TO ARBITRARY NOS.
260 FOR I%=38 TO 100
270 E(I%)=10^(-4)
280 TH1(I%)=0.
290 TH2(I%)=0.
300 NEXT I%
310 DS=204
320 PRINT TAB(0,28),"ENTER PLOTTING REQUIREMENTS VIA RED KEYS"
330 D=GET
340IF D>199 AND D<DS THEN 380 ELSE PRINTTAB(0,28),"
WRONG KEY                   "
350 FOR Q=1 TO 1500
360 NEXT
370 GOTO 320
380 II=D-201
390 IF II=-1 THEN GOTO 720
400 IF II=3 THEN 690
410 COLOUR 1:COLOUR 128
420 GCOL 0,3:GCOL 0,128
430 CLS:CLG
440 REM SET CONSTANTS FOR FRQY. INTERVAL AND OFFSET
450 DF=7./512.:OST=2./512.
460 DD=20
470 XOR=200:YOR=200:XM=800:YM=800
480 PROCBOX(XOR,YOR,XM,YM)
490 IF II=0 THEN GOTO 630
500 IF II=1 THEN GOTO 580
510 IF II=2 THEN GOTO 520
520 FL=0.05:FU=0.40
```

```
 530 PROCAXDP(XOR,YOR,DD)
 540 PROCDP(FL,FU,DF,1)
 550 PROCDP(FL,FU,DF,2)
 560 DS=205
 570 GOTO 320
 580 FL=0.05:FU=0.6
 590 PROCAXLOG(XOR,YOR,DD)
 600 PROCLOG(FL,FU,DF)
 610 DS=205
 620 GOTO 320
 630 FL=0.04:FU=0.40
 640 REM IRGE SETS VERT. SCALE ON LINEAR AXES
 650 PROCAXLIN(XOR,YOR,DD,IRGE)
 660 PROCLIN(FL,FU,DF,IRGE)
 670 DS=205
 680 GOTO 710
 690 PRINTTAB(0,28),"                                        "
 700 VDU2:VDU1,12 :CALL &C000:VDU3
 710 GOTO 320
 720 *DRIVE 0
 730 VDU3 :END
 740 ********************
 750 REM PROCEDURES USED FOR PLOTTING RESULTS
 760 REMMODIFIED FOR MIZEX CONDITIONS
 770 DEF PROCBOX(XOR,YOR,XM,YM)
 780 REM DRAWS BOX FOR GRAPH
 790 MOVE XOR,YOR
 800 DRAW XM+XOR,XOR
 810 DRAW XM+XOR,YM+YOR
 820 DRAW XOR,YM+YOR
 830 DRAW XOR,YOR
 840 ENDPROC
 850 ********************
 860 DEF PROCAXLOG(XOR,YOR,DD)
 870 REM ANNOATES LOG-LOG AXES
 880 LOCAL XK,YK,A,B,D1,D2,D3,D4
 890 XK=XOR:YK=YOR
 900 A=200:B=160
 910 D1=50:D2=40
 920 D3=130:D4=-8
 930 FOR I%=1 TO 5
 940 MOVE XK,YOR
 950 DRAW XK,YOR-DD
 960 VDU5
 970 MOVE XK-D1,YOR-D2
 980 IF I%=1 THEN PRINT ".038"
 990 IF I%=3 THEN PRINT "0.15   HZ"
1000 IF I%=5 THEN PRINT "0.6"
1010 VDU4
1020 XK=XK+A
1030 NEXT I%
1040 REM
1050 FOR I%=1 TO 6
1060 MOVE XOR,YK
1070 DRAW XOR-DD,YK
1080 VDU5
1090 MOVE XOR-D3,YK-D4
1100 IF I%=1 THEN PRINT " -4"
```

```
1110 IF I%=3 THEN PRINT " -2"
1120 IF I%=5 THEN PRINT "  0"
1130 YK=YK+B
1140 NEXT I%
1150 MOVE 80,730:PRINT "LOG"
1160 MOVE 80,650:PRINT "M*M"
1170 MOVE 67,610:PRINT " HZ"
1180 MOVE 80,630:PRINT "---"
1190 MOVE 750,950:PRINT A$
1200 MOVE 500,950:PRINT "SIR-B"
1210 VDU4
1220 ENDPROC
1230 ********************
1240 DEF PROCLOG(FL,FU,DF)
1250 REM PLOTS E ON LOG-LOG
1260 LOCAL J1%,J2%,A,B,C,D
1270 J1%=INT(FL/DF)
1280 J2%=INT(FU/DF)
1290 REM SET A & B FOR SIR-B
1300 A=1.426:B=664.5:C=4:D=160
1310 MOVE (LOG(J1%*DF+OST)+A)*B+XOR,(LOG(E(J1%))+C)*D+YOR
1320 FOR J%=J1%TO J2%
1330 X=(LOG(J%*DF+OST)+A)*B
1340 Y=(LOG(E(J%))+C)*D
1350 DRAW X+XOR,Y+YOR
1360 NEXT J%
1370 ENDPROC
1380 ********************
1390 DEF PROCAXLIN(XOR,YOR,DD,IRGE)
1400 REM AXES ON LINEAR SCALES
1410 LOCAL XK,YK,A
1420 XK=XOR:YK=YOR
1430 A=200
1440 D1=40:D2=40:D3=130:D4=-8
1450 FOR I%=1 TO 5
1460 MOVE XK,YOR
1470 DRAW XK,YOR-DD
1480 VDU5
1490 MOVE XK-D1,YOR-D2
1500 IF I%=1 THEN PRINT " 0"
1510 IF I%=3 THEN PRINT "0.2    HZ"
1520 IF I%=5 THEN PRINT "0.4"
1530 VDU4
1540 XK=XK+A
1550 NEXT I%
1560 FOR I%=1 TO 5
1570 MOVE XOR,YK
1580 DRAW XOR-DD,YK
1590 VDU5
1600 MOVE XOR-D3,YK-D4
1610 IF I%=1 THEN PRINT "  0"
1620 IF I%=3 AND IRGE=1 THEN PRINT ".05"
1630 IF I%=3 AND IRGE=2 THEN PRINT "0.5"
1640 IF I%=3 AND IRGE=3 THEN PRINT "  5"
1650 IF I%=5 AND IRGE=1 THEN PRINT "0.1"
1660 IF I%=5 AND IRGE=2 THEN PRINT "  1"
1670 IF I%=5 AND IRGE=3 THEN PRINT " 10"
1680 YK=YK+A
```

```
1690 NEXT I%
1700 MOVE 500,950:PRINT "SIR-B"
1710 MOVE 750,950:PRINT A$
1720 MOVE 80,520:PRINT "M*M"
1730 MOVE 67,480:PRINT " HZ"
1740 MOVE 80,500:PRINT "---"
1750 VDU4
1760 ENDPROC
1770 ********************
1780 DEF PROCLIN (FL,FU,DF,IRGE)
1790 REM GRAPHS ON LINEAR SCALES
1800 LOCAL J1%,J2%,A,B
1810 J1%=INT(FL/DF)
1820 J2%=INT(FU/DF)
1830 REM SET A & B FOR SIR-B
1840 A=2000
1850 REM IRGE SCALES TO 0.1,1 & 10M*M/HZ
1860 IF IRGE=1 THEN B=8000
1870 IF IRGE=2 THEN B=800
1880 IF IRGE=3 THEN B=80
1890 MOVE( J1%*DF+OST)*A+XOR,YOR
1900 FOR J%=J1% TO J2%
1910 X=(J%*DF+OST)*A
1920 Y=E(J%)*B
1930 DRAW X+XOR,Y+YOR
1940 NEXT J%
1950 ENDPROC
1960 ********************
1970DEF PROCAXDP(XOR,YOR,DD)
1980REM AXES ON LINEAR SCALES FOR DIRECTIONAL PARAMETERS
1990LOCAL XK,YK,A
2000XK=XOR:YK=YOR
2010A=200
2020D1=40:D2=40:D3=130:D4=-8
2030FOR I%=1 TO 5
2040MOVE XK,YOR
2050DRAW XK,YOR-DD
2060VDU5
2070MOVE XK-D1,YOR-D2
2080IF I%=1 THEN PRINT " 0"
2090IF I%=3 THEN PRINT "0.2    HZ"
2100IF I%=5 THEN PRINT "0.4"
2110VDU4
2120XK=XK+A
2130NEXT I%
2140FOR I%=1 TO 5
2150MOVE XOR+DD,YK
2160DRAW XOR-DD,YK
2170VDU5
2180MOVE XOR-D3,YK-D4
2190IF I%=1 THEN PRINT "180"
2200IF I%=3 THEN PRINT "  0  30"
2210IF I%=5 THEN PRINT "180  60"
2220YK=YK+A
2230NEXT I%
2240 MOVE 500,950:PRINT"SIR-B"
2250MOVE 750,950:PRINT A$
2260MOVE 80,760:PRINT"TH1"
```

```
2270MOVE 80,520:PRINT "DEG"
2280 MOVE 220,760:PRINT "TH2"
2290 MOVE 220,520:PRINT "DEG"
2300VDU4
2310ENDPROC
2320 ********************
2330DEF PROCDP(FL,FU,DF,ISEL)
2340REM PLOTS OF DIRECTIONAL PARAMETERS
2350REM ISEL=1 OR 2 FOR TH1 OR TH2
2360LOCAL J1%,J2%,A,B
2370J1%=INT(FL/DF)
2380J2%=INT(FU/DF)
2390A=2000
2400REM RANGES:360 DEG FOR TH1,60 DEG FOR TH2
2410IF ISEL=1 THEN B=800/360
2420IF ISEL=2 THEN B=800/60.
2430 IF ISEL=1 THEN YY=TH1(J1%)*B ELSE YY=TH2(J1%)*B
2440MOVE (J1%*DF+OST)*A+XOR,YY+YOR
2450FOR J%=J1% TO  J2%
2460X=(J%*DF+OST)*A
2470IF ISEL=1 THEN 2500
2480IF ISEL=2 THEN Y=TH2(J%)*B
2490 GOTO 2530
2500 Y=(TH1(J%)+180)
2510 IF Y>360 THEN Y=Y-360
2520 Y=Y*B
2530 IF ISEL=1 THEN DRAW X+XOR,Y+YOR
2540 IF ISEL=2 THEN PLOT 21,X+XOR,Y+YOR
2550NEXT J%
2560ENDPROC
2570 ********************
```



PITCH ROLL BUOY-ACORN INTERFACE FIG.1

Cross Spectral Estimates Figure 2.

FILE NAME OF INPUT GEO
FILE NAME OF OUTPUT AM
NO. OF SECTIONS 3
NO. OF HARMONICS IN SPECTRAL ESTIMATES 7

DATA VALUES

1.00	475.00	-0.03
2.00	499.00	-0.00
3.00	538.00	0.12
1.00	475.00	-0.00
2.00	499.00	0.06
3.00	538.00	-0.11
1.00	-0.03	-0.00
2.00	-0.00	0.06
3.00	0.12	-0.11

F	C11	C22	C33	Q12	Q13	C23
4.492E-2	2.326E-5	1.299E-4	1.68E-4	2.252E-5	5.816E-6	1.008E-5
5.859E-2	1.172E-4	3.067E-4	1.148E-4	1.054E-4	1.668E-5	3.372E-5
7.227E-2	1.284E-3	9.033E-4	1.095E-3	8.748E-4	7.949E-4	5.371E-4
8.594E-2	1.004E-2	9.829E-3	1.869E-3	9.647E-3	8.259E-4	7.061E-4
9.961E-2	1.012E-2	9.738E-3	3.338E-3	9.541E-3	2.512E-3	1.87E-3
0.1133	1.514E-2	1.212E-2	2.034E-3	1.317E-2	2.278E-3	1.469E-3
0.127	1.03E-2	9.685E-3	2.544E-3	9.675E-3	1.836E-3	1.536E-3
0.1406	9.89E-3	8.124E-3	3.553E-3	8.461E-3	1.003E-3	3.494E-4
0.1543	1.297E-2	1.126E-2	4.066E-3	1.16E-2	7.823E-4	2.352E-4
0.168	1.535E-2	1.195E-2	3.604E-3	1.275E-2	6.496E-4	-3.067E-4
0.1816	1.349E-2	1.246E-2	4.721E-3	1.169E-2	2.822E-3	1.306E-3
0.1953	1.495E-2	1.569E-2	6.526E-3	1.38E-2	4.278E-3	2.354E-3
0.209	1.744E-2	1.305E-2	4.006E-3	1.413E-2	4.425E-3	2.805E-3
0.2227	7.868E-3	6.643E-3	4.526E-3	6.398E-3	2.752E-3	1.85E-3
0.2363	1.036E-2	8.36E-3	4.457E-3	8.175E-3	1.122E-3	-1.371E-4
0.25	9.283E-3	6.653E-3	4.565E-3	6.54E-3	2.38E-3	6.654E-4
0.2637	8.407E-3	7.562E-3	5.215E-3	6.562E-3	1.981E-3	1.215E-4
0.2773	1.404E-2	1.272E-2	6.576E-3	1.15E-2	2.643E-3	2.388E-3
0.291	9.133E-3	7.439E-3	3.595E-3	6.808E-3	4.466E-4	-1.679E-4
0.3047	1.138E-2	1.302E-2	4.061E-3	1.065E-2	1.91E-3	1.536E-3
0.3184	1.252E-2	8.896E-3	5.881E-3	8.264E-3	1.5E-3	1.741E-3
0.332	8.398E-3	8.629E-3	9.756E-3	5.71E-3	-9.516E-4	1.926E-3
0.3457	1.163E-2	8.205E-3	1.173E-2	5.928E-3	-3.9E-3	6.91E-4
0.3594	1.72E-2	6.637E-3	8.515E-3	4.721E-3	-3.374E-3	1.36E-3
0.373	1.629E-2	1.122E-2	1.208E-2	2.79E-3	-5.352E-3	3.398E-3
0.3867	1.992E-2	7.382E-3	1.817E-2	-1.401E-3	-1.148E-2	4.001E-3
0.4004	1.934E-2	4.322E-3	1.93E-2	1.729E-3	-1.539E-2	-4.036E-4
0.4141	2.639E-2	7.482E-3	1.433E-2	1.126E-4	-1.137E-2	2.428E-3
0.4277	2.311E-2	1.067E-2	1.794E-2	-2.426E-3	-1.512E-2	3.035E-3
0.4414	1.271E-2	8.51E-3	1.283E-2	-1.165E-3	-6.667E-3	1.216E-3
0.4551	1.345E-2	1.073E-2	9.955E-3	-5.012E-3	-6.731E-3	4.597E-3
0.4687	1.358E-2	6.132E-3	1.178E-2	-6.101E-4	-9.76E-3	1.078E-3
0.4824	1.295E-2	6.921E-3	9.512E-3	-1.468E-3	-7.973E-3	1.062E-3
0.4961	9.523E-3	5.637E-3	1.447E-2	-1.803E-3	-8.333E-3	-2.616E-4
0.5098	1.607E-2	6.571E-3	7.249E-3	-4.092E-3	-6.759E-3	1.771E-3

One Dimensional Wave Spectra,Angular Harmonics
and Directional Parameters Figure 3.

F	E	A1	B1	A2	B2	R	TH1	TH12	TH2	S1	S2
0.04	0.30	0.27	0.07	-0.13	0.07	0.28	14.48	76.04	68.78	0.39	1.86
0.06	0.48	0.47	0.08	0.46	0.16	0.53	8.99	9.68	58.42	0.92	5.10
0.07	2.50	0.55	0.50	-0.10	0.54	0.80	42.26	50.05	41.48	2.82	6.20
0.09	9.70	0.89	0.08	0.68	0.12	0.93	4.89	5.03	26.45	8.39	10.38
0.10	6.12	0.83	0.22	0.49	0.29	0.88	14.75	15.15	30.57	6.02	6.63
0.11	5.21	0.90	0.16	0.71	0.21	1.03	9.81	8.12	23.84	10.55	12.97
0.13	2.19	0.86	0.16	0.58	0.25	0.92	10.75	11.64	28.38	7.15	8.40
0.14	1.43	0.79	0.09	0.39	0.06	0.92	6.76	4.35	36.88	3.83	3.95
0.15	1.39	0.82	0.06	0.47	0.03	0.92	3.86	1.87	33.97	4.69	4.92
0.17	1.10	0.83	0.04	0.54	-0.04	0.99	2.92	357.90	33.74	4.77	6.05
0.18	0.73	0.77	0.19	0.45	0.15	0.89	13.57	9.32	37.12	3.77	4.99
0.20	0.61	0.76	0.23	0.41	0.21	0.82	17.23	13.59	36.91	3.82	4.82
0.21	0.55	0.82	0.26	0.53	0.33	1.01	17.39	15.91	30.48	6.07	8.05
0.22	0.19	0.68	0.29	0.19	0.33	0.84	23.28	30.11	41.08	2.89	3.80
0.24	0.20	0.71	0.10	0.30	-0.02	0.90	7.81	357.99	43.19	2.52	3.05
0.25	0.14	0.64	0.23	0.19	0.12	0.91	20.00	16.24	45.69	2.14	2.37
0.26	0.10	0.63	0.19	0.18	0.02	0.81	16.80	2.95	47.15	1.95	2.12
0.28	0.14	0.70	0.16	0.32	0.25	0.85	12.94	18.92	43.10	2.53	4.04
0.29	0.08	0.68	0.04	0.35	-0.03	0.91	3.75	357.50	45.86	2.12	3.46
0.30	0.08	0.76	0.14	0.52	0.18	0.82	10.17	9.46	38.33	3.47	6.37
0.32	0.07	0.61	0.11	0.20	0.24	0.92	10.29	24.55	50.11	1.61	3.10
0.33	0.04	0.46	-0.08	-0.06	0.21	0.68	350.54	53.14	59.22	0.87	2.35
0.35	0.05	0.39	-0.26	-0.18	0.07	0.76	326.66	79.30	59.21	0.87	2.16
0.36	0.06	0.29	-0.21	-0.12	0.18	1.07	324.45	62.30	64.85	0.56	2.35
0.37	0.05	0.14	-0.27	-0.04	0.29	0.84	297.54	48.61	67.32	0.45	2.95
0.39	0.05	-0.06	-0.51	-0.42	0.31	0.88	263.04	71.71	56.56	1.05	5.81
0.40	0.05	0.08	-0.72	-0.63	-0.03	0.90	276.41	271.54	42.53	2.63	8.37
0.41	0.05	0.00	-0.47	-0.31	0.22	1.10	270.57	72.34	58.77	0.90	3.83
0.43	0.04	-0.09	-0.59	-0.25	0.21	0.90	260.88	70.06	51.53	1.47	3.28
0.44	0.02	-0.07	-0.40	-0.20	0.11	0.77	260.09	75.30	62.19	0.70	2.46
0.46	0.02	-0.30	-0.40	0.04	0.44	0.81	233.33	42.60	57.12	1.01	4.57
0.47	0.02	-0.04	-0.63	-0.32	0.12	0.87	266.42	79.54	49.48	1.68	3.34
0.48	0.01	-0.10	-0.55	-0.16	0.13	0.89	259.57	70.32	54.01	1.25	2.25
0.50	0.01	-0.13	-0.60	-0.44	-0.03	0.69	257.79	271.70	50.20	1.61	4.49
0.51	0.01	-0.27	-0.45	-0.05	0.26	1.08	238.81	50.40	55.54	1.13	2.68

Energy Spectra plotted on linear axes   Figure 4.



Energy Spectra plotted on logarithmic axes   Figure 5.



Plots of Mean Direction and Directional Spread Figure 6.